
Change 1 – Enhancement of IMS_PTC_DialogIndex_Type
	  type enumerated IMS_PTC_DialogIndex_Type {

    firstDialog,

    secondDialog,

    thirdDialog,

    referDialog,

    confEventDialog,
    regEventDialog,

    msgSummaryDialog,

    presenceDialog,

    xcapDiffDialogCMS,

    xcapDiffDialogGMS,
    pocSettingsDialog
  }; 


Change 2 – new function f_SIP_SUBSCRIBE_DialogIndex2EventType
	  function f_SIP_SUBSCRIBE_DialogIndex2EventType(IMS_PTC_DialogIndex_Type p_DialogIndex) return charstring

  {

    var charstring v_EventType;

    select (p_DialogIndex) {

      case (confEventDialog)   { v_EventType := "conference"; }

      case (regEventDialog)    { v_EventType := "reg"; }

      case (msgSummaryDialog)  { v_EventType := "message-summary"; }

      case (presenceDialog)    { v_EventType := "presence"; }

      case (xcapDiffDialogCMS,

            xcapDiffDialogGMS) { v_EventType := "xcap-diff"; }

      case (pocSettingsDialog) { v_EventType := "poc-settings"; }

      case else                { FatalError(__FILE__, __LINE__, "invalid dialog"); }

    }

    return v_EventType;

  }


Change 3 – fl_IMS_PTC_ImsInfo_GetDialogIndexValue
	  function fl_IMS_PTC_ImsInfo_GetDialogIndexValue(template (omit) IMS_PTC_DialogIndex_Type p_Index) runs on IMS_PTC return integer

  {

    var integer i;

    if (isvalue(p_Index)) {

      select (valueof(p_Index)) {

        case (firstDialog)      { i := 0; }

        case (secondDialog)     { i := 1; }

        case (thirdDialog,

              referDialog)      { i := 2; }

        case (confEventDialog)  { i := 3; }    /* @sic R5-153986 sic@ */

        case (regEventDialog)   { i := 4; }    /* @sic R5-153756 sic@ */

        case (msgSummaryDialog) { i := 5; }    /* @sic R5-153756, R5s150854 sic@ */

        case (presenceDialog)   { i := 6; }    /* @sic R5s190727 change 3.1 sic@ */

        case (xcapDiffDialogCMS)   { i := 7; }    /* @sic R5s190727 change 3.1 sic@ */
        case (xcapDiffDialogGMS) { i := 8; }
        case (pocSettingsDialog) { i := 9; }
        case else { FatalError(__FILE__, __LINE__, "invalid dialog"); }

      }

    } else {

      i := vc_IMS_Global.ImsInfo.Dialog.CurrentIndex;

    }

    return i;

  }


Change 4 – tsc_IMS_NoOfDialogs
	  const integer tsc_IMS_NoOfDialogs := 10;


Change 5 – a_IMS_SUBSCRIBE_Request
	  altstep a_IMS_SUBSCRIBE_Request(out IMS_DATA_REQ p_IMS_DATA_REQ,

                                  IMS_PTC_DialogIndex_Type p_DialogIndex,

                                  template (present) SipUrl p_SipUrlRequestUri,

                                  template (present) MessageHeader p_ExpectedMessageHeader,

                                  template MessageBody p_MessageBody := omit) runs on IMS_PTC

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;

    var charstring v_EventType := f_SIP_SUBSCRIBE_DialogIndex2EventType(p_DialogIndex);
    var charstring v_Tag := "SUBSCRIBE-to-" & v_EventType & "-package-" & int2char(enum2int(p_DialogIndex));
    var charstring v_ToTag := f_IMS_GenerateTag(v_Tag);    /* @sic R5-153756: dialog for subscribe sic@ */


    [] a_IMS_ReceiveRequest(car_IMS_Subscribe_Request(cr_SUBSCRIBE_Request(p_SipUrlRequestUri, p_ExpectedMessageHeader, p_MessageBody)), v_IMS_DATA_REQ_ByRef)

      {

        v_MessageHeader := v_IMS_DATA_REQ_ByRef.Request.Subscribe.msgHeader;

        v_Protocol := v_IMS_DATA_REQ_ByRef.RoutingInfo.Protocol;

        f_IMS_MessageHeader_CheckTo(v_MessageHeader, noToTag);           // This cannot be checked in the template

        f_IMS_MessageHeader_Request_CheckVia(v_MessageHeader, v_Protocol);      // @sic R5s120727 change 9 sic@

        f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);                  // @sic R5s120727 change 9 sic@

        f_IMS_MessageHeader_Request_CheckRoute(v_MessageHeader);                // @sic R5s140934 change 2 sic@

        if (f_IMS_PTC_ImsInfo_SigCompStarted()) {

          f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

        }









        f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_MessageHeader, p_DialogIndex);               /* @sic R5-153756: dialog for subscribe sic@ */

        f_IMS_PTC_ImsInfo_DialogSetLocalTag(v_ToTag, p_DialogIndex);                              /* @sic R5-153756: dialog for subscribe sic@ */

        p_IMS_DATA_REQ := v_IMS_DATA_REQ_ByRef;

      }

  }

	


Change 6 – f_IMS_SubscribeRequest_MessageHeaderRX_C30

	  function f_IMS_SubscribeRequest_MessageHeaderRX_C30(IMS_PTC_DialogIndex_Type p_DialogIndex) runs on IMS_PTC return template (present) MessageHeader

  { /* @sic R5-155913: From/To-header acc. to dialog being created by initial subscribtion sic@ */

    /* @sic R5s160045: CSeqValue shall not be incremeted in this function (i.e. f_IMS_RequestInDialog_CSeqValueRX shall not be used) as function is used in altesteps what would cause incrementing of CSeqValue for every snapshot - independent from whether or not there is a match sic@ */

    /* @sic R5-174544: SessionId removed sic@ */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var EventType v_EventType := f_SIP_SUBSCRIBE_DialogIndex2EventType(p_DialogIndex);

    var boolean v_CondA5 := f_IMS_PTC_SecurityScheme_IsSipDigest();  // FBBA

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());









    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");   /* @sic R5s160045: any CSeqValue; CSeqValue to be check after the SUBSCRIBE has been effectively matched an appropriate receive statement sic@ */

    v_MessageHeader_Subscribe.fromField   := f_IMS_RequestInDialog_FromHeaderRX(p_DialogIndex);             // @sic R5s150843: p_DialogIndex sic@

    v_MessageHeader_Subscribe.toField     := f_IMS_RequestInDialog_ToHeaderRX(p_DialogIndex);               // @sic R5s150843: p_DialogIndex sic@

    v_MessageHeader_Subscribe.callId      := f_IMS_PTC_ImsInfo_DialogGetCallId(p_DialogIndex);              // @sic R5s150843: p_DialogIndex sic@

    v_MessageHeader_Subscribe.expires     := cr_Expires("0");

    v_MessageHeader_Subscribe.event       := cr_EventDef(v_EventType);

    if (v_CondA5) {   // FBBA   @sic R5-168943 FBBA  un-subscribe sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo;

    }

    return v_MessageHeader_Subscribe;

  }


Change 7 – f_IMS_NotifyRequest_MessageHeader_C30

	  function f_IMS_NotifyRequest_MessageHeader_C30(IMS_PTC_DialogIndex_Type p_DialogIndex,

                                                 SUBSCRIBE_Request p_SubscribeRequest) runs on IMS_PTC return template (value) MessageHeader

  {

    var charstring v_NW_AddrStr := f_IMS_PTC_NW_Address_GetAddrStr();

    var CharStringList_Type v_ViaAddressList := {v_NW_AddrStr, tsc_IMS_Scscf};

    var template (value) SubscriptionState v_SubscriptionState := cs_SubscriptionState_Terminated;

    var template (omit) ContentType v_ContentType := omit;

    var template (value) SipUrl v_ContactSipUrl;

    var template (value) Via v_Via;

    var EventType v_EventType := f_SIP_SUBSCRIBE_DialogIndex2EventType(p_DialogIndex);

    select (p_DialogIndex) {

      case (regEventDialog)  {        // A3


        v_ContactSipUrl := cs_SipUri_HostPort(tsc_IMS_Scscf);

      }

      case (confEventDialog) {        // A4


        v_ContactSipUrl := p_SubscribeRequest.requestLine.requestUri;    /* the UE shall use the FinalConferenceUri in the request line of the SUBSCRIBE message */

      }

      case (msgSummaryDialog) {       // A5


        /* v_ViaAddressList[lengthof(v_ViaAddressList)] := f_IMS_PTC_ImsInfo_GetHomeDomainName();  !!!! PROSE ISSUE: acc. to A.6.2 the via header for message-summary has 3 entries  !!!!

         *                                                                                         !!!!              nevertheless C.30 step 0C defines only 2 entries !!!! */

        v_ContactSipUrl := cs_SipUri_HostPort(tsc_IMS_Scscf);

      }

      case (presenceDialog) {


        // v_ContactSipUrl := cs_SipUri_HostPort(tsc_IMS_Scscf);

        FatalError(__FILE__, __LINE__, "not supported yet");

      }

      case (xcapDiffDialogCMS, xcapDiffDialogGMS) {


        // v_ContactSipUrl := cs_SipUri_HostPort(tsc_IMS_Scscf);

        FatalError(__FILE__, __LINE__, "not supported yet");

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid dialog");

      }

    }

    v_Via := f_IMS_NotifyRequest_ViaTX(v_ViaAddressList);

    return f_IMS_NotifyRequest_MessageHeader_Common(p_DialogIndex, v_Via, cs_Contact(v_ContactSipUrl), v_EventType, v_SubscriptionState, v_ContentType);

  }


Change 8 – f_IMS_SUBSCRIBE_Request
	  function f_IMS_SUBSCRIBE_Request(IMS_PTC_DialogIndex_Type p_DialogIndex := regEventDialog,

                                   
                                   template SipUrl p_SipUrlTo := omit,

                                   template (present) charstring p_ExpireTime := "600000") runs on IMS_PTC return IMS_DATA_REQ

  { /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;

    var EventType v_EventType := f_SIP_SUBSCRIBE_DialogIndex2EventType(p_DialogIndex);
    var template (present) SipUrl v_SipUrlRequestUri := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) MessageHeader v_ExpectedMessageHeader := f_IMS_SubscribeRequest_MessageHeaderRX(v_EventType, p_SipUrlTo, p_ExpireTime);

    a_IMS_SUBSCRIBE_Request(v_IMS_DATA_REQ_ByRef, p_DialogIndex , v_SipUrlRequestUri, v_ExpectedMessageHeader);       /* @sic R5-150705: altstep a_IMS_SUBSCRIBE_Request sic@

                                                                                                                      @sic R5s200000: v_ExpectedMessageHeader sic@ */

    return v_IMS_DATA_REQ_ByRef;

  }


Change  – f_IMS_Subscription_MWI_Event_Package
	  function f_IMS_Subscription_MWI_Event_Package() runs on IMS_PTC

  {

    var REGISTER_Request v_RegisterRequest;

    var IMS_DATA_REQ v_IMS_REGISTER_REQ;

    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;

    var SUBSCRIBE_Request v_SubscribeReq_MsgSummaryEvent;

    var CharStringList_Type v_IMPU_List;

    var SipUrl v_ContactUrl_RegEvent;

    var template (present) MessageHeader v_ExpectedMessageHeaderMsgSummary;

    var template (present) MessageHeader v_ExpectedMessageHeaderRegEvent;

    var template (present) SipUrl v_RequestUrl_RegEvent;

    var template (present) SipUrl v_RequestUrl_MsgSummaryEvent;

    var template (present) SipUrl v_ToUrl_MsgSummaryEvent;

    var charstring v_MessageServerDomain;

    var Opt_msg_headers v_SetOfOptionalLines;

    var template (present) charstring v_ExpireTime := ?;   // value not to be checked acc. to A.6.1

    var boolean v_SubcribedToRegEventPackage := false;

    var boolean v_SubcribedToMsgSummaryEventPackage := false;

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    v_IMS_REGISTER_REQ := f_IMS_Register();

    v_RegisterRequest := v_IMS_REGISTER_REQ.Request.Register;

    v_RequestUrl_RegEvent := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    v_ContactUrl_RegEvent := f_MessageHeader_GetContactSipUrl(v_RegisterRequest.msgHeader);

    v_RequestUrl_MsgSummaryEvent := cr_SipUri_HostPort(?);        // to be checked after SUBSCRIBE has been received

    v_ToUrl_MsgSummaryEvent := cr_SipUri_HostPort(?);             // to be checked after SUBSCRIBE has been received

    v_MessageServerDomain := f_IMS_PTC_ImsInfo_GetHomeDomainName();

    f_IMS_TestBody_Set(true);

    v_ExpectedMessageHeaderMsgSummary := f_IMS_SubscribeRequest_MessageHeaderRX("message-summary", v_ToUrl_MsgSummaryEvent, v_ExpireTime);

    v_ExpectedMessageHeaderRegEvent := f_IMS_SubscribeRequest_MessageHeaderRX("reg");

    alt {

      // Step 1. UE subscribes to the Message Waiting Indication event package.

      [not v_SubcribedToMsgSummaryEventPackage] a_IMS_SUBSCRIBE_Request(v_IMS_DATA_REQ_ByRef, msgSummaryDialog, v_RequestUrl_MsgSummaryEvent, v_ExpectedMessageHeaderMsgSummary) {

        v_SubscribeReq_MsgSummaryEvent := v_IMS_DATA_REQ_ByRef.Request.Subscribe;

        if (px_IMS_MessageAccountIdentity != "") {

          v_IMPU_List := { px_IMS_MessageAccountIdentity };      /* UE shall use MessageAccountIdentity when configured */

        } else {

          v_IMPU_List := f_IMS_PTC_ImsInfo_GetAllIMPUs();        /* any IMPU within the set of IMPUs on ISIM if MessageAccountIdentity is not configured at the UE */

        }

        if (not f_IMS_PublicUserId_Match(v_IMPU_List, v_SubscribeReq_MsgSummaryEvent.requestLine.requestUri)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid request URI");

        }

        if (not f_IMS_PublicUserId_Match(v_IMPU_List, f_Addr_Union_GetSipUrl(v_SubscribeReq_MsgSummaryEvent.msgHeader.toField.addressField))) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid request URI");

        }

        // Step 2. The SS responds SUBSCRIBE with 200 OK

        f_IMS_SUBSCRIBE_Response(msgSummaryDialog, v_IMS_DATA_REQ_ByRef, "7200");   /* @sic R5-155913: ExpireTime := "7200" sic@ */

        v_SubcribedToMsgSummaryEventPackage := true;

        if (not v_SubcribedToRegEventPackage) { repeat; }

      }

      // Step 2a. The UE subscribes to the registration event package.

      [not v_SubcribedToRegEventPackage] a_IMS_SUBSCRIBE_Request(v_IMS_DATA_REQ_ByRef, regEventDialog, v_RequestUrl_RegEvent, v_ExpectedMessageHeaderRegEvent) {

        // Step 2b. The SS responds with 200 OK

        f_IMS_SUBSCRIBE_Response(regEventDialog, v_IMS_DATA_REQ_ByRef);

        v_SubcribedToRegEventPackage := true;

        if (not v_SubcribedToMsgSummaryEventPackage) { repeat; }

      }

    }

    // @siclog "Step 3 - 4" siclog@

    fl_IMS_Notify_MWI(v_SubscribeReq_MsgSummaryEvent, v_MessageServerDomain);

    // @siclog "Step 5 - 6" siclog@

    v_SetOfOptionalLines := {

      "To: " & f_SIP_SipUrl2Str(f_Addr_Union_GetSipUrl(v_SubscribeReq_MsgSummaryEvent.msgHeader.fromField.addressField)),

      "From: <user2_public1@home1.net>",

      "Subject: call me back!",

      "Message-ID: 27775334485@" & v_MessageServerDomain,

      "Message-Context: voice-message",

      "Date: Fri 09 Dec 2016 09:15 +0100",        // @sic R5-168947 sic@

      "Priority: urgent"                          // @sic R5-168947 sic@

    };

    fl_IMS_Notify_MWI(v_SubscribeReq_MsgSummaryEvent, v_MessageServerDomain, v_SetOfOptionalLines);

    // @siclog "Step 7 - 8" siclog@

    f_IMS_Register_Notify(v_ContactUrl_RegEvent, f_IMS_XmlMessageForNotify_Def(v_RegisterRequest));    /* @sic R5-153756, R5-153986: p_SubscribeRequest and p_CseqValue removed sic@ */

    f_IMS_TestBody_Set(false);

  }


